**********************
*Purged export shocks*
**********************
*
set more off
use temp1,clear
egen naf2yr = group(naf2 year)
global depvar "llaborpty"
global depvar_d1  "llaborpty_d1"
label var shock3intensd  "log trade shock\_intens--purged"
label var shock6intensd  "$\tilde{\Delta}$ trade shock\_intens--purged"
*
****************************
*FIRST TABLE: REDUCED FORM *
****************************
*NAIVE
*FE
eststo: xtreg  $depvar shock3intensd  i.naf2yr, robust cluster(firmid) fe i(firmid)
*FD
eststo: reg 	$depvar_d1  shock6intensd i.naf2yr, robust cluster(firmid)
*FE+FD
eststo: xtreg   $depvar_d1  shock6intensd  i.naf2yr, robust cluster(firmid)  fe i(firmid)
*INV AND MATERIALS CONTROLED
*FE
eststo: xtreg  $depvar shock3intensd linv lmatraw i.naf2yr, robust cluster(firmid) fe i(firmid)
*FD
eststo: reg 	$depvar_d1  shock6intensd linv_d1 lmatraw_d1 i.naf2yr, robust cluster(firmid)
*FE+FD
eststo: xtreg   $depvar_d1  shock6intensd linv_d1 lmatraw_d1 i.naf2yr, robust cluster(firmid)  fe i(firmid)
*
log using $DIRLOG/Replication/tab-ReducedForm-purged.txt, text replace
*
set linesize 255
****************************************************************************
*FIRST TABLE: REDUCED FORM: NAIVE + innovation and materials purged schocks*
****************************************************************************
esttab, keep(shock* linv* lmatraw*) b(%4.3f) se(%4.3f) r2 starlevels({$^c$} 0.1 {$^b$} 0.05 {$^a$} 0.01)  tex  title() label
esttab, keep(shock* linv* lmatraw*) compress r2 starlevels(c 0.1 b 0.05 a 0.01)  b(%4.3f) se(%4.3f)
*output the results in a tex file directly inputable in the text
esttab using $DIRLOG/Replication/tab-ReducedForm-purged_input.tex, b(%4.3f) se(%4.3f)  starlevels({$^c$} 0.1 {$^b$} 0.05 {$^a$} 0.01)    ///
nocons keep(shock* linv* lmatraw*)  label tex substitute(\_ \_)  fragment replace nodepvars nonumbers nomtitles nogaps nolines eqlabels(none)
eststo clear
corr shock3intens shock3intensd
corr shock6intens shock6intensd
*
log close


